
OPC XML-DA Client

Requisiti

Per il funzionamento del client è necessario installare la versione 1.1 del Framework .NET. .NET necessita di un sistema operativo Windows NT 4.0 o Windows 98 o versioni successive.

Descrizione dell’interfaccia

Finestra principale

La finestra principale è divisa in 4 parti:

1. albero dei gruppi

2. lista degli item appartenenti al gruppo selezionato (indicato a destra nella barra di stato), tutti o una parte nel caso in cui sia attivo un filtro. Facendo click col tasto destro su questa lista appare un menu pop-up che permette di copiare o modificare gli item

3. lista delle proprietà dell’item selezionato. E’ aggiornata da un click o dalla pressione di Invio su un item

4. lista dei messaggi. Visualizza tra l’altro gli errori e il risultato della chiamata di GetStatus

Menu della finestra principale

Server

· Seleziona web service... [image: image12.png]=




Permette la selezione del web service. Ha lo stesso effetto del pulsante “Server”
· [image: image13.png]=



Stato
Chiama la primitiva GetStatus Ha lo stesso effetto del pulsante “GetStatus”
· Opzioni...
Permette di cambiare il LocaleID (stringa di localizzazione) e il carattere che separa i branch all’interno del percorso degli item
· Chiudi
Esce dal programma
Item

· Filtro…  (CTRL-F)
Permette di cambiare il filtro per i nomi degli item da visualizzare
· Copia (CTRL-C)
Copia gli item selezionati. Successivamente si potranno incollare in una lista di lettura, scrittura o sottoscrizione
· Seleziona tutti  (CTRL-A)
Seleziona tutti gli item
· Modifica…  (CTRL-E)
Visualizza la finestra “Modifica item” che permette di cambiare valore, e, se supportato, timestamp e quality dell’item selezionato 
Finestra

· Refresh

· Gruppi[image: image14.png]=




Aggiorna l’albero dei gruppi. Ha lo stesso effetto del pulsante “Refresh”
· Items (F5)

· Aggiorna la lista degli item
· Proprietà (F6)
Aggiorna la lista delle proprietà
· Nuova reading list[image: image15.png]



Crea una finestra che permette la lettura contemporanea di più item
· Nuova subscription list
Crea una finestra che permette l’accesso al servizio di sottoscrizione
· Nuova write list
Crea una finestra che permette la scrittura contemporanea di più item
Il menu visualizza la lista delle finestre create

Scelta del server

La finestra OPC XML-DA Web Service consente la selezione del web service mediante l’inserimento dell’URL. Gli ultimi URL inseriti saranno memorizzati e disponibili successivamente anche dopo il riavvio del programma. All’avvio, il programma tenta di connettersi con l’ultimo web service usato. Per ogni server è importante scegliere il corretto carattere di separazione dei branch all’interno dei path (che di default è il punto) nella finestra “Opzioni”.

Finestra Modifica item

La finestra “Modifica item” è accessibile da tutte le liste di item (browser, read list, write list, subscription list) mediante il comando “Modifica” o “Modifica immediata” del menu pop-up).

Finestra Read list

La finestra “Read list” permette la lettura contemporanea di più item mediante la primitiva Read. Esistono due modi per costruire una lista di item: la copia di item da un'altra lista e l’inserimento manuale del percorso degli item. Per eseguire quest’ultima operazione si deve premere il pulsante nella parte destra della finestra “Aggiungi un nuovo item” e poi modificare opportunamente il contenuto della casella di testo “Item”. Per incollare item copiati da altre liste bisogna fare click col tasto destro del mouse in un punto qualsiasi della lista, apparirà il seguente menu contestuale:

Successivamente si deve selezionare “Incolla”. Altre voci del menù permettono di selezionare tutti gli elementi della lista, eliminare o copiare gli elementi selezionati e modificare il valore di un singolo item.

“MaxAge” indica la freshness dei dati richiesta espressa in ms, se il valore contenuto in cache è più vecchio di quanto indicato da questo paramentro deve essere riletto dal device. MaxAge pari a 0 indica che il dato deve essere letto direttamente dal device. La primitiva Read permette opzionalmente di specificare un valore di MaxAge diverso per ogni item, ma questo client non lo consente. Dopo aver impostato MaxaAge si può premere il pulsante “Read”: se ci sono errori (ad esempio percorso sconosciuto o item senza diritto di accesso in lettura) il valore di ResultID corrispondente verrà visualizzato nella colonna “Errori”, una descrizione estesa dell’errore sarà visualizzata nella lista dei messaggi della finestra principale.

Il tasto “<<” torna alla finestra principale.

Finestra Subscribe

La finestra Subscribe permette l’accesso al servizio di sottoscrizione. La prima operazione da effettuare è la costruzione della lista degli item. Questa operazione è analoga a quella vista per la lettura di liste di item, con la differenza che per ogni item si possono specificare il rate di campionamento richiesto al server (RequestedSamplingRate), la Deadband e se deve essere abilitato il buffering (EnableBuffering). Se specificati, questi parametri sono prioritari rispetto a quelli inseriti nelle caselle della parte superiore della finestra (i valori di default, vedi Subscribe) e per modificarli bisogna cambiare il contenuto delle caselle di testo e di controllo nella parte destra della finestra dopo aver selezionato un item. Il pulsante “Subscribe” invoca la primitiva Subscribe, una volta premuto sulla caption bar apparirà il valore restituito dal server come handle per la sottoscrizione ed eventualmente i RequestedSamplingRate saranno corretti in base ai valori supportati dal server.

Si può modificare la lista, ma solo dopo aver premuto il pulsante “SuscriptionCancel”, che invoca la primitiva SuscriptionCancel. Questo perché non è possibile modificare una sottoscrizione, ma se ne deve creare un’altra (che sarà identificata da un nuovo handle).

Dopo aver premuto il pulsante Subscribe si possono effettuare i polling dei dati. Si selezioni la scheda “Polling”. Il pulsante “SubscriptionPolledRefresh” invoca la primitiva SubscriptionPolledRefresh. Si può osservare come per gli item per i quali era stato richiesto il buffering delle variazioni si hanno più voci restituite se tra una pressione e l’altra del pulsante il RequestedSamplingRate scade più di una volta.

Il significato dell’hold time e del wait time è spiegato in Meccanismo di sottoscrizione. L’hold time passato alla primitiva SubscriptionPolledRefresh è un tempo assoluto, il valore inserito sarà pari all’intervallo tra la pressione del pulsante “SubscriptionPolledRefresh” e l’hold time.

I valori ritornati vengono accodati alla lista, per cancellare il contenuto della lista bisogna premere il pulsante “Pulisci finestra”.

Dopo aver premuto il pulsante “Polling ciclico” questo rimane schiacciato e si avvia un thread che invoca continuamente la primitiva SubscriptionPolledRefresh finchè non si preme nuovamente il pulsante. In questo caso l’hold time non può essere inferiore ad un secondo.

Finestra Write list

La finestra “Write list” permette la scrittura contemporanea di più item mediante la primitiva Write. L’aspetto e il funzionamento della finestra è molto simile a quello della Read list, la differenza è che per ogni item le proprietà che appaiono sulla destra della finestra sono modificabili. Oltre ai valori si possono inviare le informazioni sulla qualità e il timestamp. La scrittura di qualità e timestamp può non essere supportata dal server, che in questo caso (se i campi sono stati specificati) dovrebbe rispondere con un errore E_NOTSUPPORTED, senza che questo pregiudichi la scrittura del valore. Se le caselle relative alle informazioni opzionali non vengono riempite, questi campi non verranno inviati al server.

Codice

Il client è stato realizzato con il Microsoft Visual Basic .NET (Microsoft Visual Studio .NET 2003). L’interfacciamento con il server è possibile grazie alla classe OpcXmlDa. Questa è una classe proxy, cioè una classe che gestisce il mapping dei parametri delle primitive con gli elementi XML e il conseguente invio del messaggio SOAP in rete. In pratica è uno stub, ma di serializzazione e deserializzazione dei messaggi si occupano le librerie del Framework .NET. In particolare è stato necessario aggiungere il riferimento allo spazio dei nomi System.Web.Services (Progetto>Aggiungi riferimento…>.NET> System.Web.Services.dll). La classe è stata generata automanticamene a partire dal file WSDL grazie al tool wsdl.exe del Microsoft .NET Framework SDK:

wsdl XMLDA10.wsdl /language:VB /out:OpcXmlDa.vb

XMLDA10.wsdl è la descrizione del web service, on line all’indirizzo: http://opcfoundation.org/webservices/XMLDA/1.0/

VB indica Visual Basic .NET, il linguaggio in cui sarà generata la classe, alternative sono CS e JS, che indicano rispettivamente C# e JScript .NET

Nel modulo global.vb è definita una istanza globale della classe OpcXmlDa:

Public Srv As OpcXmlDa = New OpcXmlDa

All’avvio del programma e ogni volta che si seleziona il server viene impostata la proprietà Url di tale oggetto.

Finestra principale (formMain.vb)

Il codice della finestra principale esemplifica l’utilizzo delle primitive GetStatus, Browse e GetProperties.

All’avvio del programma, dopo la scelta del server o alla selezione della voce del menu Finestra>Refresh>Gruppi viene aggiornato il contenuto del controllo tvBranches di tipo TreeView mediante l’esecuzione della sub refreshBranches():

Public Sub refreshBranches()

    Dim newNode As New System.Windows.Forms.TreeNode

    Me.Cursor = Cursors.WaitCursor 'fa apparire la clessidra
    tvBranches.Nodes.Clear()       'cancella l'albero esistente
    newNode.Text = "Server"

    makeTree(newNode, "")

    tvBranches.Nodes.Add(newNode)  'crea il nodo principale
    lvItems.Items.Clear()

    lvProperties.Items.Clear()

    Me.Cursor = Cursors.Default

End Sub

Public Sub makeTree(ByRef node As System.Windows.Forms.TreeNode, ByVal path As String)

    Dim Items() As BrowseElement

    Dim Errors() As OPCError

    Dim MoreElements As Boolean

    Dim reply As ReplyBase

    Dim ContPoint As String

    Dim prop() As System.Xml.XmlQualifiedName

    Try

        reply = Srv.Browse(prop, LocaleID, getClientHandle(), "", path, ContPoint, MAX_ELEMENTS_RET, browseFilter.branch, "", "", False, False, True, Items, Errors, MoreElements)

        'stampa errori e descrizioni
        For Each err As OPCError In Errors

            addToLog(err.ID.Name + "    " + err.Text)

        Next

        Dim bi As BrowseElement

        For Each bi In Items

            Dim newNode As New System.Windows.Forms.TreeNode

            newNode.Text = bi.Name

            'bi.Name dovrebbe essere uguale all'ultima parte di bi.ItemName:

            'paths = bi.ItemName.Split(BranchSepChar)

            'newNode.Text = paths(paths.Length - 1)

            'e sarà il testo visualizzato sul controllo

            If bi.HasChildren Then makeTree(newNode, bi.ItemName)

            node.Nodes.Add(newNode)

        Next

        If MoreElements Then addToLog("Esistono più di " + MAX_ELEMENTS_RET + " elementi")

    Catch err As Exception

        addToLog(err.Message)

    End Try

End Sub
La riga più importante è l’invocazione del metodo Browse dell’oggetto Srv:

reply = Srv.Browse(prop, LocaleID, getClientHandle(), "", path, ContPoint, MAX_ELEMENTS_RET, browseFilter.branch, "", "", False, False, True, Items, Errors, MoreElements)

Il path è il parametro della sub makeTree richiamata ricorsivamente, ed è il percorso del ramo del quale si vogliono leggere gli elementi contenuti. Il parametro browseFilter è impostato a “browseFilter.branch” per fare in modo che vengano restituiti solo i rami (e non gli item semplici), i tre parametri booleani False, False e True indicano rispettivamente che non devono essere restituite tutte le proprietà ma solo quelle indicate nell’array prop (che comunque è vuoto), che non devono essere restituiti i valori delle proprietà indicate e che dovrà essere ritornato l’attributo Text (la descrizione) degli errori che con la riga:

addToLog(err.ID.Name + "    " + err.Text)

viene visualizzato nella parte bassa della finestra.

I punti di continuazione non vengono usati perché non supportati da tutti i server. Per evitare che qualche item non sia visualizzato bisogna impostare a 0 la costante MAX_ELEMENTS_RET definita in global.vb (vedi Browse). Per usare i punti di continuazione dovremmo racchiudere il codice in un loop:

Do

    ...

Loop While MoreElements

Quando viene selezionata una voce all’interno dell’albero o alla selezione della voce del menu Finestra>Refresh>Items viene invocata la sub refreshItems() per aggiornare il controllo lvItems di tipo ListView:

Private Sub refreshItems()

    Dim Items() As BrowseElement

    Dim Errors() As OPCError

    Dim MoreElements As Boolean

    Dim reply As ReplyBase

    Dim ContPoint As String

    Dim path As String = getCurrentPath()

    Dim clientHandle = getClientHandle()

    Me.Cursor = Cursors.WaitCursor

    Dim prop() As System.Xml.XmlQualifiedName

    lvItems.Items.Clear()

    lvProperties.Items.Clear()

    Try

        reply = Srv.Browse(prop, LocaleID, clientHandle, "", path, ContPoint, MAX_ELEMENTS_RET, browseFilter.item, ItemNameFilter, "", False, False, True, Items, Errors, MoreElements)

        For Each err As OPCError In Errors

            addToLog(err.ID.Name + "    " + err.Text)

        Next

        Dim bi As BrowseElement

        For Each bi In Items

            lvItems.Items.Add(bi.Name, 2)

        Next

        If MoreElements Then addToLog("Esistono più di " + MAX_ELEMENTS_RET + " elementi")

    Catch err As Exception

        addToLog(err.Message)

    End Try

    Me.Cursor = Cursors.Default

End Sub

Il codice è simile a quello della sub makeTree ma non è ricorsivo, la primitiva Browse viene invocata con browseFilter =browseFilter.item e viene impostato il filtro ItemNameFilter come parametro BrowseFilter, che in makeTree era una stringa vuota.

Alla selezione di un elemento della lista di proprietà, alla pressione di Invio quando lvItems è il controllo attivo o alla selezione del comando del menù Finestra>Refresh>Proprietà la sub refreshProperties() aggiorna la lista delle proprietà dell’item selezionato:

Private Sub refreshProperties()

    lvProperties.Items.Clear()

    If lvItems.SelectedItems.Count <> 1 Then Exit Sub

    Dim iId(1) As ItemIdentifier

    iId(0) = New ItemIdentifier

    Dim prop() As System.Xml.XmlQualifiedName

    Dim propReply() As PropertyReplyList

    Dim errors() As OPCError

    Dim reply As ReplyBase

    iId(0).ItemName = getCurrentItem()

    Try

        reply = Srv.GetProperties(iId, prop, LocaleID, getClientHandle(), "", True, True, True, propReply, errors)

        For Each err As OPCError In errors

            addToLog(err.ID.Name + "    " + err.Text)

        Next

        Dim pl As PropertyReplyList

        For Each pl In propReply

            If Not IsNothing(pl.Properties) Then

                For i As Integer = 0 To pl.Properties.GetLength(0) - 1

                    Dim newProp As New ListViewItem

                    newProp.SubItems.Add("")

                    newProp.SubItems.Add("")

                    newProp.SubItems.Add("")

                    If Not IsNothing(pl.Properties(i).Value) Then newProp.SubItems(1).Text = pl.Properties(i).Value.ToString

                    If Not IsNothing(pl.Properties(i).ResultID) Then newProp.SubItems(2).Text = pl.Properties(i).ResultID.Name

                    If Not IsNothing(pl.Properties(i).Description) Then

                        newProp.SubItems(0).Text = pl.Properties(i).Description

                    Else

                        newProp.SubItems(0).Text = isntNothing(pl.Properties(i).Name)

                    End If

                    lvProperties.Items.Add(newProp)

                Next i

            End If

        Next pl

        Dim newItem As ListViewItem

    Catch err As Exception

        addToLog(err.Message)

    End Try

End Sub

Come la primitiva Browse, la primitiva GetProperties permette di ottenere gli attributi di tutte le proprietà o solo di quelle indicate in un array (prop). Col 6° parametro (True) scegliamo la prima alternativa, col 7° (True) indichiamo che vogliamo ottenere anche i valori delle proprietà. Osserviamo che l’array iId è riempito con un solo valore (il percorso dell’item attualmente selezionato) per cui, se non ci sono errori, il valore restituito propReply sarà un array di un solo elemento (pl=propReply(0)). Quando esiste la descrizione estesa della proprietà (attributo Descriptiom) verrà visualizzata questa, altrimenti verrà visualizzato il nome (attributo Name).

Alla selezione del comando Stato del menù Server verrà eseguita la seguente sub:

Public Sub callGetStatus()

    Dim reply As ReplyBase

    Dim status As ServerStatus

    Dim SendTime As Date = Now()

    Try

        reply = Srv.GetStatus(LocaleID, getClientHandle(), status)

        Dim li As String

        If status.SupportedLocaleIDs.Length Then

            For Each lid As String In status.SupportedLocaleIDs

                If li <> "" Then li += ",  "

                li += lid

            Next

        End If

        Dim iv As String = ""

        If status.SupportedInterfaceVersions.Length Then

            For Each ivd As interfaceVersion In status.SupportedInterfaceVersions

                If iv <> "" Then iv += ",  "

                iv += ivd.ToString()

            Next

        End If

        Dim lst As String

        lst = "Server avviato alle: " + timeToString(status.StartTime)

        Dim rss As String

        rss = "Server State: " + reply.ServerState.ToString()

        Dim tel As String

        tel = "Tempo trascorso dalla richiesta: " & Now.Subtract(SendTime).TotalSeconds & " secondi"

        addToLog(Chr(10) + "Vendors: " + status.VendorInfo + Chr(10) + _

            "Product Version: " + status.ProductVersion + Chr(10) + _

            "Status Info: " + status.StatusInfo + Chr(10) + _

            "Interface Version: " + iv + Chr(10) + _

            "Locale Ids: " + li + Chr(10) + _

            lst + Chr(10) + rss + Chr(10) + _

            "Revised Locale Id: " + reply.RevisedLocaleID + Chr(10) + tel + Chr(10))

    Catch err As Exception

        addToLog(err.Message)

    End Try

End Sub

Finestra Read list (formRead.vb)

Alla pressione del pulsante “Read” viene eseguito il seguente codice:

Private Sub buttRead_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles buttRead.Click

    If GItemList1.lvItems.Items.Count = 0 Then

        MsgBox("Deve essere presente almeno un elemento nella lista", MsgBoxStyle.Exclamation)

        Exit Sub

    End If

    Dim reply As New ReplyBase

    Dim rOpt As New RequestOptions

    Dim items As New ReadRequestItemList

    Dim ritems As ReplyItemList

    Dim errors As OPCError()

    rOpt.ReturnErrorText = True

    rOpt.ReturnItemTime = True

    rOpt.ClientRequestHandle = getClientHandle()

    rOpt.LocaleID = LocaleID

    If txtMaxAge.Text = "" Then

        items.MaxAgeSpecified = False

    Else '(se la casella di testo non è vuota)
        items.MaxAge = CInt(txtMaxAge.Text)

        items.MaxAgeSpecified = True

    End If

    ReDim items.Items(GItemList1.lvItems.Items.Count)

'il seg. ciclo riempie l'array items.Items con i percorsi degli item con l'aiuto della funzione joinItemName, che costruisce il nome "fully qualified" a partire da percorso e nome corto
    For x As Integer = 0 To GItemList1.lvItems.Items.Count - 1

        items.Items(x) = New ReadRequestItem

        items.Items(x).ItemName = joinItemName(GItemList1.lvItems.Items(x).SubItems(0).Text, GItemList1.lvItems.Items(x).SubItems(1).Text)

    Next x

    Try

        reply = Srv.Read(rOpt, items, ritems, errors)

        For Each err As OPCError In errors

            formMainI.addToLog(err.ID.Name + "    " + err.Text)

        Next

'il seg. ciclo aggiorna la ListView in base al contenuto dei valori restituiti dal metodo Read, contenuti nell'array ritems.Items
        For x As Integer = 0 To GItemList1.lvItems.Items.Count - 1

            With GItemList1.lvItems.Items(x)

                If Not IsNothing(ritems.Items(x).ResultID) Then .SubItems(5).Text = ritems.Items(x).ResultID.Name Else .SubItems(5).Text = ""

                If Not ritems.Items(x).Value.GetType.IsArray Then .SubItems(2).Text = ritems.Items(x).Value

                .SubItems(6).Text() = isntNothing(ritems.Items(x).ValueTypeQualifier)

                .SubItems(4).Text = isntNothing(ritems.Items(x).Timestamp)

                If IsNothing(ritems.Items(x).Quality) Then

                    .SubItems(7).Text = ""

                    .SubItems(3).Text = ""

                    .SubItems(8).Text = ""

                Else

                    .SubItems(7).Text = GItemList1.cbLimitField.Items(ritems.Items(x).Quality.LimitField)

                    .SubItems(3).Text = GItemList1.cbQualityField.Items(ritems.Items(x).Quality.QualityField)

                    .SubItems(8).Text = isntNothing(ritems.Items(x).Quality.VendorField)

                End If

            End With

        Next x

    Catch err As Exception

        MsgBox(err.Message, MsgBoxStyle.Exclamation)

    End Try

End Sub

Finestra Subscribe (formSubscribe.vb)

Alla pressione del tasto Subscribe sarà invocato il metodo Subscribe dell’oggetto Srv:

reply = Srv.Subscribe(rOpt, items, False, 600000, subReplyItemList, errors, serverSubHandle)

600000 è il tempo massimo in ms (10 minuti) che può trascorrere senza una richiesta da parte del client senza che il server lo consideri assente e liberi le risorse associate alla sottoscrizione;

items è un array che viene riempito dall’esecuzione del seg. codice per ogni elemento della lista:

With GItemList1.lvItems.Items(x)

    items.Items(x) = New SubscribeRequestItem

    items.Items(x).ItemName = joinItemName(.SubItems(0).Text, .SubItems(1).Text)

    copyIfcond(items.Items(x).RequestedSamplingRate, .SubItems(2).Text, items.Items(x).RequestedSamplingRateSpecified)

    copyIfcond(items.Items(x).Deadband, .SubItems(3).Text, items.Items(x).DeadbandSpecified)

    If .SubItems(4).Text <> "" Then

        items.Items(x).EnableBuffering = (.SubItems(4).Text = "si")

        items.Items(x).EnableBufferingSpecified = True

    Else

        items.Items(x).EnableBufferingSpecified = False

    End If

With

copyIfcond() riempie un campi opzionali se il 2° parametro non è una stringa vuota.

Se alcuni SamplingRate non sono supportati, il server lo notifica:

For x As Integer = 0 To subReplyItemList.Items.Length – 1

With GItemList1.lvItems.Items(x)

If Not IsNothing(subReplyItemList.Items(x).ItemValue.ResultID) Then .SubItems(5).Text = subReplyItemList.Items(x).ItemValue.ResultID.Name Else .SubItems(5).Text = ""

If subReplyItemList.Items(x).RevisedSamplingRateSpecified Then .SubItems(2).Text = subReplyItemList.Items(x).RevisedSamplingRate


End With
Next x

If subReplyItemList.RevisedSamplingRateSpecified Then txtSamplingRate.Text = subReplyItemList.RevisedSamplingRate

clockOffset è una variabile membro di formSubscribe che contiene lo sfasamento tra il clock del server e il clock del client, il calcolo di tale sfasamento è effettuato dopo l’invocazione di Subscribe:

sndTime = DateAndTime.Timer

reply = Srv.Subscribe(...)

rcvTime = DateAndTime.Timer

'il calcolo tiene conto di un tempo di propagazione stimato come la metà del round trip time corretto sottraendo il tempo di processamento del server (cioè reply.ReplyTime - reply.RcvTime)
propagationTime = (rcvTime - sndTime - _

                   reply.ReplyTime.Subtract(reply.RcvTime).TotalSeconds) / 2

clockOffset = Date2SecOfDay(reply.RcvTime) - sndTime - propagationTime

Dim dayDiff As Integer = DateAndTime.DateDiff(DateInterval.Day, Today, reply.RcvTime.Date)

'la condizione è imposta solo per evitare overflow
If Math.Abs(dayDiff) < 24 Then clockOffset += dayDiff * 86400

Il pulsante “SubscriptionPolledRefresh” invoca Srv.SubscriptionPolledRefresh:

HoldTime = Now.AddMilliseconds(txtHoldTime.Text  + clockOffset * 1000)

Srv.SubscriptionPolledRefresh(rOpt, hnd, HoldTime, True, txtWaitTime.Text, False, tmp, ritems, errors, ovflow)

Il pulsante “Polling ciclico” crea un thread che invoca continuamente lo stesso metodo, con alcune differenze:

Srv.SubscriptionPolledRefresh(rOpt, hnd, HoldTime, Not firstTime, txtWaitTime.Text, firstTime, tmp, ritems, errors, ovflow)

Si osservi che i parametri HoldTimeSpecified e ReturnAllItems sono rispettivamente posti a firstTime e Not firstTime dove firstTime è una variable booleana che assume valore true solo alla prima esecuzione del ciclo. In pratica, la prima volta vengono richiesti tutti gli item per riempire la lista e viene specificato di ignorare i parametri HoldTime e WaitTime in modo da ottenere immediatamente (o meglio il più presto possibile) i valori degli item. Nei cicli successivi viene invece utilizzato il meccanismo di sottoscrizione avanzato e verranno restituiti soltanto gli item modificati dall’ultima chiamata. Il client si occuperà di individuare quali sono gli elementi della lista che devono essere aggiornati.

La cancellazione della sottoscrizione (pulsante “SubscriptionCancel”) richiede la più semplice delle primitive:

Srv.SubscriptionCancel(serverSubHandle, getClientHandle())

La funzione getClientHandle() restituisce una stringa sempre diversa che viene utilizzato come handle delle diverse richieste, in realtà questo client non ha bisogno di controllare gli handle ritornati e il parametro ClientHandle potrebbe essere sempre una stringa nulla.

Finestra Write list (formWrite.vb)

Parte del codice eseguito alla pressione del pulsante “Write” è il seguente:

ReDim items.Items(GItemList1.lvItems.Items.Count)

Dim x As Integer

For Each li As ListViewItem In GItemList1.lvItems.Items

    items.Items(x) = New ItemValue

    With items.Items(x)

        .Value = li.SubItems(2).Text

        If li.SubItems(6).Text <> "" Then .ValueTypeQualifier = New System.Xml.XmlQualifiedName(li.SubItems(6).Text)

        .ItemName = joinItemName(li.SubItems(0).Text, li.SubItems(1).Text)

        .Quality = New OPCQuality

        copyIf2(.Quality.LimitField, GItemList1.cbLimitField.Items.IndexOf(li.SubItems(7).Text))

        copyIf2(.Quality.QualityField, GItemList1.cbLimitField.Items.IndexOf(li.SubItems(3).Text))

        copyIf(.Quality.VendorField, li.SubItems(8).Text)

        copyIfcond(.Timestamp, li.SubItems(4).Text, .TimestampSpecified)

    End With

    x += 1

Next li

rOpt.ReturnErrorText = True

rOpt.ClientRequestHandle = getClientHandle()

rOpt.LocaleID = LocaleID

reply = Srv.Write(rOpt, items, False, ritems, errors)

Finestra Modifica item (formEdit.vb)

Il codice richiamato alla visualizzazione della finestra “Modifica item” è simile a quello eseguito alla pressione del tasto “Read” nella finestra “Read list”, mentre quello associato all’evento buttOk.Click è simile al codice eseguito alla pressione del tasto “Write” nella finestra “Write list”. In entrambi i casi, però, verrà letto o creato un array (items.Items) di un solo valore.

Links

· .NET Framework Version 1.1 Redistributable Package (necessario per l’esecuzione del client):
http://www.microsoft.com/downloads/details.aspx?FamilyId=262D25E3-F589-4842-8157-034D1E7CF3A3&displaylang=it
La pagina elenca i requisiti aggiuntivi per una installazione server

· Server XML-DA presenti on-line:

· http://tswinc.us/XMLDADemo/XML_Sim/OpcXmlDaServer.asmx

· http://tswinc.us/XMLDADemo/ts_sim/OpcDaGateway.asmx

· http://tswinc.us/DemoUDAWeb/UDAWeb.asmx
(quest’ultimo usa “/” come carattere di separazione tra i branch all’interno dei path, è necessario settarlo selezionando il comando “Opzioni…” dal menu “Server” del client affinché il browser funzioni correttamente)

· XDARap Rapid Server Toolkit: http://www.tswinc.us/showproduct.aspx?Productid=12&categoryid=9. Contiene i server XML-DA di esempio usati per testare il client. Preventivamente è necessario installare Internet Information Services (IIS) versione 5.0 o successiva (quindi non è possibile usare Windows 98 o ME) e, qualora non siano già presenti nel sistema, gli altri componenti necessari all’istallazione server del FrameWork .NET


· OPC Foundation: http://www.opcfoundation.org/
· Simple Object Access Protocol: http://www.w3.org/TR/SOAP/
· XML-Data (il formato dei file wsdl): http://www.w3.org/TR/1998/NOTE-XML-data/

